Le théorème des cordes sécantes avec Maple

Le théorème des cordes sécantes se trouve dans les éléments d'Euclide. Il se résume dans le schéma suivant :



Pour le démontrer par les complexes, on place l'origine du plan complexe au centre du cercle. Les affixes des points A,B,C,D sont donc supposées être des complexes de même module (que l'on peut supposer être égaux à 1). On note a,b,c et d ces affixes. Leurs conjugués sont donc leur inverse. On note p l'affixe de P. On commence par calculer p et son conjugué.

```
> eq1:=expand((p-a)*(conjugate(p)-1/b)-(p-b)*(conjugate(p)-1/a));
```

$$eq1 := -\frac{p}{b} - a\,\bar{p} + \frac{a}{b} + \frac{p}{a} + b\,\bar{p} - \frac{b}{a}$$

Cette égalité traduit l'alignement de A,P et B. On procéde de même avec C,P et D

```
> eq2:=subs({a=c,b=d},eq1);
```

$$eq2 := -\frac{p}{d} - c\,\bar{p} + \frac{c}{d} + \frac{p}{c} + d\,\bar{p} - \frac{d}{c}$$

```
> sol:=solve({eq1,eq2},{p,conjugate(p)});
```

$$sol := \{p = \frac{b\,a\,d + b\,a\,c - b\,d\,c - a\,d\,c}{b\,a - d\,c}, \bar{p} = \frac{b - d - c + a}{b\,a - d\,c}\}$$

On a donc l'affixe de p. Reste à calcule (PA x PB)²

```
> PAPB_carre:=subs(sol,(p-a)*(p-b)*(conjugate(p)-1/a)*(conjugate(p
)-1/b));
```

$$PAPB\_carre := \left(\frac{b\,a\,d + b\,a\,c - b\,d\,c - a\,d\,c}{b\,a - d\,c} - a\right)\left(\frac{b - d - c + a}{b\,a - d\,c} - \frac{1}{a}\right)$$
$$\left(\frac{b\,a\,d + b\,a\,c - b\,d\,c - a\,d\,c}{b\,a - d\,c} - b\right)\left(\frac{b - d - c + a}{b\,a - d\,c} - \frac{1}{b}\right)$$

```
> factor(PAPB_carre);
```

$$\frac{(a-c)^2\,(d-a)^2\,(b-c)^2\,(b-d)^2}{(b\,a-d\,c)^4}$$

L'invariance par "a donne c", "b donne d" prouve que (PA x PB)²=(PC x PD)², ce que l'on peut vérifier par

```
> PCPD_carre:=subs(sol,(p-c)*(p-d)*(conjugate(p)-1/c)*(conjugate(p
)-1/d));
```

$$PCPD\_carre := \left(-c + \frac{b\,a\,d + b\,a\,c - b\,d\,c - a\,d\,c}{b\,a - d\,c}\right)\left(\frac{b - d - c + a}{b\,a - d\,c} - \frac{1}{c}\right)$$
$$\left(-d + \frac{b\,a\,d + b\,a\,c - b\,d\,c - a\,d\,c}{b\,a - d\,c}\right)\left(\frac{b - d - c + a}{b\,a - d\,c} - \frac{1}{d}\right)$$

```
> factor(PCPD_carre);
```

$$\frac{(a-c)^2\,(d-a)^2\,(b-c)^2\,(b-d)^2}{(b\,a-d\,c)^4}$$

```
>
```